

データの検索（２）

エクセルでは、「一覧表からデータを検索してくる」という使い方をする場面が多くあります。今回は、**Index** 関数を使ってみます。また、**Index** 関数と相性のいい **Match** 関数も、あわせて使ってみます。

旅行カタログによく出ている、見にくい料金表を例に、関数を使った料金早見表を作ってみましょう。

1. Index 関数・Match 関数の基礎

	A	B	C	D	E	F	G	H
1			1	2	3		商品	3
2			A店	B店	C店		お店	1
3	1	赤いきつね	90	87	94		値段	97
4	2	どん兵衛	88	87	105			
5	3	チャルメラ	97	87	105			
6	4	UFO	100	128	145			
7								

Index(範囲, 行数, 列数, 検索方法)

という書式です。

たとえば、上図 A1～E6 表のように、A～C 店で、各商品の値段がいくらかを作成します。H1 セルに商品の行位置、H2 セルにお店の列位置を入力したときに、H3 セルに自動的に値段を出したいとします。そのときは、

=Index(C3:E6, H1, H2)

となります。

「商品の名前から、何番目のデータなのかを知る」場合は、**Match** 関数を使います。

Match(検索値, 範囲, 照合の種類)

「照合の種類」には-1, 0, 1 のいずれかが入りますが、通常は 0（完全一致）を使うものと思っていたほうがいいでしょう。

I1 セルに商品名を入力した場合、H1 セルに、それが B3～B6 の何番目のデータかを表示するには、

=Match(I1,B3:B6,0)

とします。

同様に、H2 セルに I2 のお店が C2～E2 の何番目のデータかを表示されるように、関数を入力してみましょう。

	G	H	I
商品		3	チャルメラ
お店		1	A店
値段		97	

2. 準備

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z	AA	AB	AC	AD	AE	AF
1	出発日カレンダー																															
2																																
3	月\日	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26	27	28	29	30	31
4	4月	C	C	C	A	A	A	A	B	B	A	A	A	A	A	B	B	A	A	A	A	A	B	B	A	A	A	A	C	C	-	
5	5月	C	C	B	A	B	B	A	A	A	A	A	B	B	A	A	A	A	B	B	A	A	A	A	A	B	B	A	A	A	-	
6	6月	A	B	B	A	A	A	A	A	B	B	A	A	A	A	A	B	B	A	A	A	A	C	C	B	B	B	A	A	B	-	

「出発日」シートを作り、上図のような、出発日カレンダーを作ってください。

次に、「料金表」シートを作り、右図のような、パッケージツアーによくある料金表を作ってください。

	A	B	C	D
1	料金表 (4泊5日、2名1室利用)			
2				
3	ドージョー・ビーチホテル			
4	部屋タイプ	A	B	C
5	グランドタワー	74,900	78,900	83,900
6	オーシャンタワー	77,900	80,900	85,900
7	ドージョークラブ	94,900	98,900	105,900
8	ドージョースイート	116,900	120,900	139,900

最後に、「早見表」シートを作って、右図のような表を作ってください。

[問]

(1) B4セルは、「出発日」シートのA4～A6のリストから選べるように、リスト設定してください。

(2) C4セルは、「出発日」シートのB3～AF3のリストから選べるように、リスト設定してください。

(3) B5セルは、「料金表」シートのA5～A8のリストから選べるように、リスト設定してください。

(4) B6セルに、出発日カレンダーの該当する記号が自動的に入るように、関数を入力してください。

(5) B7セルに、上セル内容で料金がいくらかになるのか、を自動的に表示する関数を入力してください。

	A	B	C	D
1	料金早見表			
2				
3	ホテル名	ドージョー・ビーチホテル		
4	出発日	4月	5日	
5	部屋タイプ	ドージョークラブ		
6	出発日記号			
7	料金			

	A	B	C	D
1	料金早見表			
2				
3	ホテル名	ドージョー・ビーチホテル		
4	出発日	4月	5日	
5	部屋タイプ	ドージョークラブ		
6	出発日記号	A		
7	料金			94,900

[解答]

(1) (2) (3) は、「データ」タブ→「データの入力規則」で、「リスト」を選びます。「元の値」は、該当するデータ範囲を入力してください。

(4) =Index(出発日!B4:AF6, Match(B4,出発日!A4:A6,0), Match(C4,出発日!B3:AF3,0))

(5) =Index(料金表!B5:D8, Match(B5,料金表!A5:A8,0), Match(B6,料金表!B4:D4,0))